



WORKING IN SGPLOT


Understanding the General
Logic of Attributes

Graphical Elements in SGPLOT

- All graphs generated by SGPLOT can be viewed as a collection of elements.
- Some of the nomenclature of these elements is the same or similar to that which was utilized in SAS/GRAPH.
 - Labels on axes, along with values at major tick marks.
 - Legends still have values for each category, but in SGPLOT we have a title instead of a label.
 - In scatterplots, and related graphs, we have markers—the keyword symbol is used to refer to their shape.
- Any of the elements can potentially be re-styled by altering their attributes.
 - In SGPLOT, graphical elements are divided into various types that have common attribute sets.

Two Simple Plots—Bar Chart


```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    hbar origin / response=mpg_city  
        stat=mean;  
run;  
quit;
```



Two Simple Plots—Bar Chart

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    hbar origin / response=mpg_city  
        stat=mean;  
  
run;  
quit;
```

Labels and values on any axis are text graphical elements. For text, 5 attributes can be set:




Two Simple Plots—Bar Chart

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    hbar origin / response=mpg_city  
        stat=mean;  
  
run;  
quit;
```

Labels and values on any axis are text elements. For text, 5 attributes can be set:


Font, size, color, style and weight



Two Simple Plots—Bar Chart

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    hbar origin / response=mpg_city  
        stat=mean;  
  
run;  
quit;
```


In their default form, bars have two elements—fill and outline. Fills and lines each have their own attribute sets.



Two Simple Plots—Bar Chart


```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    hbar origin / response=mpg_city  
        stat=mean;  
run;  
quit;
```

Fill allows for color and transparency.
Line allows color, pattern and thickness.



Two Simple Plots—Scatterplot


```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    scatter x=horsepower y=mpg_city;  
    where type ne 'Hybrid';  
run;  
quit;
```



Two Simple Plots—Scatterplot

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    scatter x=horsepower y=mpg_city;  
    where type ne 'Hybrid';  
run;  
quit;
```


One element the bar graph does not have
is the marker, which the scatterplot
contains by default.



Two Simple Plots—Scatterplot

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
    scatter x=horsepower y=mpg_city;  
    where type ne 'Hybrid';  
run;  
quit;
```

For markers, color, size and symbol can all be set.




The Fill Attribute

- We will start by modifying the bar fill in the first graph.
- Fill attributes available for modification are:
 - Color—in any of the SAS color models
 - Transparency—a proportion, from 0 to 1, with 0 being fully opaque
- Since the hbar statement generates the bars, these attributes can be set as an option in the hbar statement, in general:
 - The keyword is derived from the element name with the suffix ***attrs***
 - For the bar fill, we will use ***fillattrs=(options)***
 - It is also possible to assign a style element as part of this specification—beyond the scope of this lesson

The Fill Attribute

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city  
    stat=mean  
    fillattrs=(color=cxFFAA00);  
run;  
quit;
```

Add *fillattrs* to your list of hbar options and choose your favorite color.



The Fill Attribute

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city  
  stat=mean  
  fillattrs=(c=cxFFAA00);  
run;  
quit;
```


If you are used to SAS/GRAPH,
common aliases do not work here.

```
Log - (Untitled)  
55  proc sgplot data=sashelp.cars;  
56      hbar origin / response=mpg_city stat=mean  
57      fillattrs=(c=cxFFAA00);  
                                     -  
                                     22  
                                     202  
ERROR 22-322: Syntax error, expecting one of the following: COLOR,  
TRANSPARENCY.  
ERROR 202-322: The option or parameter is not recognized and will be  
ignored.  
58  run;  
  
NOTE: The SAS System stopped processing this step because of errors.  
NOTE: PROCEDURE SGPLOT used (Total process time):  
      real time           0.12 seconds  
      cpu time            0.01 seconds  
  
59  quit;
```

The Fill Attribute

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city  
    stat=mean  
    fillattrs=(color=cxFFAA00  
               transparency=0.5);  
run;  
quit;
```


With no overlays of graphs,
transparency amounts to little more
than a lightening of the color.



The Fill Attribute

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean;  
  hbar origin / response=mpg_highway stat=mean  
    fillattrs=(transparency=0.4  
               color=yellow);  
run;  
quit;
```


Here, yellow bars for highway MPG overlay the default blue bars for city. The transparency effect results in a green for the city bars.



The Fill Attribute

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean;  
  hbar origin / response=mpg_highway stat=mean  
    fillattrs=(transparency=0.4 color=cxFF6666)  
    barwidth=0.6;  
run;  
quit;
```

Some care with colors and bar sizes
can be helpful in these situations.




Text Attributes

- Returning to our simple bar graph, we will alter text styles.
- Text attributes available for modification are:
 - Color—in any of the SAS color models
 - Family—any font available in your session
 - Size—in various units: cm, mm, in, pct, pt, px
 - Style—normal or italic
 - Weight—normal or bold
- The hbar statement generates text only due to the axes:
 - The only text modifications we can make to the original graph will be in an ***xaxis*** or ***yaxis*** statement.
 - The keyword is still derived from the element name with the suffix ***attrs***
 - For the labels, we will use ***labelattrs=(options)***—for the values, we will use ***valueattrs=(options)***

Text Attributes


```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean;  
  yaxis display=(nolabel)  
    valueattrs=(size=14pt style=italic);  
  xaxis labelattrs=(family='Arial Black');  
run;  
quit;
```

So in each axis statement we have modified one of the text elements.




Text Attributes

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg city stat=mean;  
  yaxis display=(nolabel)  
    valueattrs=(size=14pt style=italic);  
  xaxis labelattrs=(family='Arial Black');  
run;  
quit;
```



Text Attributes

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean;  
  yaxis display=(nolabel)  
    valueattrs=(size=14pt style=italic);  
  xaxis labelattrs=(family='Arial Black');  
run;  
quit;
```




Exercise

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean  
    datalabel;  
run;  
quit;
```

Of course, text may appear in the graph area for a bar graph, such as when the ***datalabel*** option is used.

Can you modify the attributes of the data labels?




Exercise

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean  
    datalabel datalabelattrs=(family='Georgia'  
                               size=12pt);  
run;  
quit;
```

Following the rules, the attribute keyword must be ***datalabelattrs*** (yes they can get long) and it must be an option in the graphing statement.

Any option that generates a graphic element can be reliably linked to its ***attrs*** keyword.




Line Attributes

- Returning to our simple bar graph, we will alter the bar outlines.
- Line attributes available for modification are:
 - Color—in any of the SAS color models
 - Pattern—numbers 1 to 46 for various patterns (or names for named patterns)
 - Thickness—in various units: cm, mm, in, pct, pt, px
- Since the hbar statement generates the bars with outlines (by default), these attributes can be set as an option in the hbar statement :
 - The keyword is still derived from the element name with the suffix ***attrs***
 - So we will use ***outlineattrs=(options)***

Line Attributes

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean  
    outlineattrs=(color=cxFFAA55 thickness=2pt);  
run;  
quit;
```

Knowing the names of default graph elements is helpful, as it gives a direct link to the *attrs* keyword.





Marker Attributes

- The final type of graph element associated with attributes are markers—we will use the scatterplot to look at these.
- Marker attributes available for modification are:
 - Color—in any of the SAS color models
 - Size—in various units: cm, mm, in, pct, pt, px
 - Symbol—various shapes, see next slide for choices
- The scatter statement generates a marker at each data point:
 - The keyword is still derived from the element name with the suffix ***attrs***
 - So we will use ***markerattrs=(options)***—as an option in the scatter statement



Marker Symbols


Supported Marker Symbols

ArrowDown	↓	StarFilled	★
Asterisk	✱	Tack	⌞
Circle	○	Tilde	~
CircleFilled	●	Triangle	△
Diamond	◇	TriangleFilled	▲
DiamondFilled	◆	TriangleDown	▽
GreaterThan	>	TriangleDownFilled	▼
Hash	#	TriangleLeft	◁
HomeDown	⬇	TriangleLeftFilled	◀
HomeDownFilled	⬇	TriangleRight	▷
IBeam	⌵	TriangleRightFilled	▶
LessThan	<	Union	∪
Plus	+	X	×
Square	□	Y	Y
SquareFilled	■	Z	Z
Star	☆		

Marker Attributes

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  scatter x=horsepower y=mpg_city/  
    markerattrs=(symbol=squarefilled  
                color=cxAA6666 size=4pt);  
  where type ne 'Hybrid';  
run;  
quit;
```

Whatever is chosen is
applied to all markers.




Marker Attributes—A Special Case

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  scatter x=horsepower y=mpg_city/filledoutlinedmarkers  
    markerattrs=(symbol=squarefilled  
                 color=cxAA6666 size=4pt);  
  where type ne 'Hybrid';  
run;  
quit;
```

The option ***filledoutlinedmarkers*** (yes these can be long) gives a peculiar result with our previous attributes.

The shape is correct (a filled symbol is required for ***filledoutlinedmarkers*** to work) and the size is also. However, neither the fill nor the outline use the red.


Can you figure out why and modify these to have a fill of a light blue and an outline in the red we used?



Marker Attributes—A Special Case


```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  scatter x=horsepower y=mpg_city/filledoutlinedmarkers  
    markerattrs=(symbol=squarefilled size=4pt)  
    markerfillattrs=(color=blue)  
    markeroutlineattrs=(color=cxAA6666);  
  where type ne 'Hybrid';  
run;  
quit;
```

With *filledoutlinedmarkers*, the marker fill and outline become separate elements. Realizing this, then it follows that each gets its own attrs keyword.




Groups

```
ods graphics / reset;
proc sgplot data=sashelp.cars;
  hbar origin / response=mpg_city stat=mean
    group=type groupdisplay=cluster;
  where type not in ('Truck', 'Hybrid');
run;
quit;
```




```
ods graphics / reset;
proc sgplot data=sashelp.cars;
  scatter x=horsepower y=mpg_city/
    group=origin;
  where type ne 'Hybrid';
run;
quit;
```




Groups

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  hbar origin / response=mpg_city stat=mean  
    group=type groupdisplay=cluster;  
  where type not in ('Truck','Hybrid');  
run;  
quit;
```




```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  scatter x=horsepower y=mpg_city/  
    group=origin;  
  where type ne 'Hybrid';  
run;  
quit;
```



styleattrs Statement

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  styleattrs datacolors=(cx66FF66 cxFF6666 cx6666FF cxFFAA66);  
  hbar origin / response=mpg_city stat=mean  
    group=type groupdisplay=cluster;  
  where type not in ('Truck','Hybrid');  
run;  
quit;
```


The *styleattrs* statement can be used to set lists of values to replace the default lists SAS cycles through.



styleattrs Statement

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  styleattrs datacolors=(cx66FF66 cxFF6666 cx6666FF cxFFAA66);  
  hbar origin / response=mpg_city stat=mean  
    group=type groupdisplay=cluster;  
  where type not in ('Truck','Hybrid');  
run;  
quit;
```


The *datacolors* option lets you set the list of fill colors.



styleattrs Statement

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  styleattrs datasymbols=(trianglefilled squarefilled starfilled);  
  scatter x=horsepower y=mpg_city/group=origin;  
  where type ne 'Hybrid';  
run;  
quit;
```


The *datasymbols* option lets you set the list of marker shapes.



styleattrs Statement

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  styleattrs datasymbols=(trianglefilled squarefilled starfilled)  
    datacontrastcolors=(red green blue);  
  scatter x=horsepower y=mpg_city/  
    group=origin;  
  where type ne 'Hybrid';  
run;  
quit;
```

The *datacontrastcolors* option lets you set the list of marker colors (and lines, if applicable).






Styleattrs Statement

- Also includes:
 - ***datalinepatterns***—patterns for line elements
 - ***backcolor***—background color of the graph
 - ***wallcolor***—color for the plot wall
 - Maintenance release 3 and above for the color options

Symbol and Color Priority for Markers

```
ods graphics / reset;  
proc sgplot data=sashelp.cars;  
  styleattrs datasymbols=(trianglefilled squarefilled)  
  datacontrastcolors=(red green);  
  scatter x=horsepower y=mpg_city/  
    group=origin;  
  where type ne 'Hybrid';  
run;  
quit;
```

Because of the default behavior of changing color and symbol simultaneously, the four combinations that could be created from 2 colors and 2 shapes are not used effectively by default.



attrpriority Option in ODS Graphics

```
ods graphics / reset attrpriority=color;  
proc sgplot data=sashelp.cars;  
  styleattrs datasymbols=(trianglefilled squarefilled)  
  datacontrastcolors=(red green);  
  scatter x=horsepower y=mpg_city/  
    group=origin;  
  where type ne 'Hybrid';  
run;  
quit;
```

The default setting for *attrpriority* is none, which can be changed only to color. This causes SAS to cycle through colors prior to cycling through symbols.

